
Declarative mining of negative sequential patterns
Philippe Besnard 1 and Thomas Guyet 2
Abstract. Declarative pattern mining consists in using declarative
frameworks to solve pattern mining tasks. In this article, we address
the task of mining negative sequential patterns in Answer Set Programming (ASP). A negative sequential pattern is specified by means
of a sequence consisting of events to occur and of other events, called
negative events, to be absent. For instance, containment of the pattern
ha ¬b ci arises with an occurrence of a and a subsequent occurrence
of c but no occurrence of b in between. Recent results shed light on
the ambiguity of such a seemingly intuitive notation, exhibited three
semantics of the negative events and proposed alternative notations
for them. In this article, we propose Answer Set Programming encodings of these three semantics in order to extract frequent negative
sequential patterns from a set of sequences. It relies on previous encodings of frequent sequential pattern mining. We experiment with
our encoding on synthetic data and compare the numbers of extracted
patterns and the computing time obtained for each kind of negation.
Surprisingly, the semantics that has the best algorithmic properties
for pattern mining is not associated to an encoding that is the most
efficient.

1

Introduction

Pattern mining [11] is a data mining task that consists in extracting interesting structured patterns from a set of structured examples.
The standard pattern mining tasks are itemset mining, sequence mining and graph mining. The interestingness measure of a pattern is, in
most of the algorithms, the number of its occurrences in the set of examples. Given a threshold k, interesting patterns are those that occur
at least in k examples. In this case, the task is known as frequent pattern mining for which many algorithms have been proposed. Most of
the efficient algorithmic solutions rely on an antimonotonicity property of the support: the larger the pattern, the fewer it occurs.
Declarative pattern mining aims at encoding pattern tasks in a
declarative framework, and more specifically the frequent pattern
mining tasks. Declarative pattern mining addressed the tasks of
frequent itemset mining [5, 13, 12], frequent sequential patterns
[15, 2, 1] or frequent graph mining [6]. Different declarative frameworks have been explored: Constraint Programming (CP) [5, 15, 6],
SAT [12, 1] and Answer Set Programming (ASP) [13, 2]. Declarative
pattern mining does not expect to be competitive with dedicated algorithms, but to take advantage of the versatility of declarative frameworks to propose pattern mining tools that could exploit background
knowledge during the mining process to extract less but meaningful
patterns.
In this work, we explore negative sequential patterns [8, 10, 17].
Standard sequential pattern mining algorithms [14] extract sequential patterns that frequently occur in a set of sequences. A sequential

pattern is a sequence of events or items. For example, the sequential
pattern ha c di is read as “a occurs and then c occurs and finally d
occurs”. Negative sequential patterns are sequential patterns that also
specify non-occurring events. Intuitively, the syntax of a simple negative sequential pattern is as follows: ha ¬b ci. This pattern is read
as “a occurs and then c occurs, but b does not occur in between”. A
negative sequential pattern can also be the premise of a rule. Guyet
and Besnard [8] highlighted that ¬ symbol has several semantics but
only few of them have good properties for pattern mining.
In this article, we propose an ASP encoding of the mining of frequent negative sequential patterns. Our encoding relies on previous
work on frequent sequential pattern mining in ASP [9]. The objective
of this article is also to illustrate the semantics of negative sequential
patterns through declarative encodings.

2
2.1

Background
Sequential pattern mining

Our terminology on sequence mining follows the one in [15].
Throughout this article, [n] = {1, . . . , n} denotes the set of the first
n positive integers.
Let Σ be the set of items (alphabet). An itemset A =
{a1 , a2 , · · · , am } ⊆ Σ is a finite set of items. The size of A, denoted |A|, is m. A sequence s is of the form s = hs1 s2 · · · sn i
where each si is an itemset. n is the length of the sequence.
A database D is a multiset of sequences over Σ.A sequence s =
hs1 . . . sm i with si ∈ Σ is contained in a sequence t = ht1 . . . tn i
with m ≤ n, written s v t, if si ⊆ tei for 1 ≤ i ≤ m and an
increasing sequence (e1 . . . em ) of positive integers ei ∈ [n], called
an embedding of s in t. For example, we have ha (cd)i v ha b (cde)i
relative to embedding (1, 3). (cd) denotes the itemset made of items
c and d.
Given a database D, the cover of a sequence p is the set of sequences in D that contain p: cover (p, D) = {t ∈ D | p v t}.
The number of sequences in D containing p is called its support,
that is, support(p, D) = |cover (p, D)|. For an integer k, frequent sequence mining is about discovering all sequences p such that
support(p, D) ≥ k. We often call p a (sequential) pattern, and k is
also referred to as the (minimum) frequency threshold.

2.2

Answer set programming

A logic program is a set of rules of the form
a0 :- a1 ; . . . ; am ; not am+1 ; . . . ; not an .

(1)

where each ai is a propositional atom for 0 ≤ i ≤ n and not
stands for default negation. If n = 0, rule (1) is called a fact. If
a0 is omitted, (1) represents an integrity constraint. Semantically, a

seq(1,1,d).
seq(2,1,a).
seq(3,1,a).
seq(4,1,a).
seq(5,1,a).
seq(6,1,b).
seq(7,1,c).

seq(1,2,a).
seq(2,2,c).
seq(3,2,b).
seq(4,1,b).
seq(5,2,c).

seq(1,2,b). seq(1,3,c).
seq(2,3,b). seq(2,4,c).
seq(3,2,c). seq(3,3,a).
seq(4,2,c).

Listing 1. Facts specifying a database of sequences

logic program induces a collection of so-called answer sets, which
are distinguished models of the program determined by answer sets
semantics; see [4] for details. To facilitate the use of ASP in practice,
several extensions have been developed. First of all, rules with variables are viewed as shorthands for the set of their ground instances.
Further language constructs include conditional literals and cardinality constraints [16]. The former are of the form a : b1 , . . . , bm , the
latter can be written as s {c1 , . . . , cn } t, where a and bi are possibly
default negated literals and each cj is a conditional literal; s and t
provide lower and upper bounds on the number of satisfied literals
in a cardinality constraint. The practical value of both constructs becomes more apparent when used in conjunction with variables. For
instance, a conditional literal like a(X) : b(X) in a rule’s antecedent
expands to the conjunction of all instances of a(X) for which the
corresponding instance of b(X) holds. Similarly, 2 {a(X) : b(X)} 4
holds whenever between two and four instances of a(X) (subject to
b(X)) are true. Specifically, we rely in the sequel on the input language of the ASP system clingo [3].

3

ASP-based Sequence Mining

In this section, we introduce the sequential pattern mining tasks in
ASP. The following programs correspond to the notion of skip-gaps
encodings in [2, 7]. Gebser et al. [2] uses a different encoding strategy for frequent sequential pattern that is not suitable for negative
patterns.

3.1

Fact format

We represent a database D in terms of facts seq(t,p,e), saying
that item e occurs at position p in a sequence t. For instance, Listing 1
specifies a database of seven sequences:
Term
1
2
3
4
5 6 7
Seq. hd (ab) ci ha c b ci ha (bc) ai h(ab) ci ha ci hbi hci

Considering the minimal frequency threshold k = 2, one can
check that hai, hbi, hci, ha bi, h(ab)i, ha ci, hb ci, and h(ab) ci are
the frequent patterns of the database.

3.2

Mining frequent sequences: basic encoding

The encoding principle for frequent sequence mining follows the one
of [13], that is, each answer set comprises a single pattern of interest.
In contrast to itemset mining, however, we need to take the order of
items in a pattern into account to determine its support and check its
frequency.
Listing 2 provides our basic encoding of frequent sequence mining borrowed from [9]. It relies on two parameters: max determines a
maximum length for patterns of interest, and k specifies the minimal
frequency threshold. An answer set then represents a frequent pattern p = hp1 . . . pm i such that m ∈ [max] by atoms pat(m,p1 ),

1 item(I) :- seq(T,P,I).
3 slot(1).
4 { slot(X+1) } :- slot(X); X<max.
5 1{ pat(X,E) : item(E) } :- slot(X).
7 patlen(L) :- pat(L,_); not pat(L+1,E):item(E).
9 emb(T,P,1) :- seq(T,P,E):pat(1,E); seq(T,P,_).
10 emb(T,P,X+1) :- emb(T,Q,X); seq(T,P,_); Q<P;
11
seq(T,P,E):pat(X+1,E);
12
patlen(L); X<L.
14 cover(T) :- patlen(L); emb(T,_,L).
16 :- not k { cover(T) }.
Listing 2. Basic encoding of frequent sequence mining

. . . , pat(1,pm ). That is, the first argument expresses the positions of items. For instance, the atoms pat(1,a), pat(2,b), and
pat(2,c) stand for the (frequent) pattern ha (bc)i of the database
given in Listing 1.
A complete specification of atoms used in our ASP encodings
along with their meanings is given in Table 1. In more detail, the rule
in Line 1 of Listing 2 provides items occurring in a given database D.
Lines 3 and 4 provides the slots for the itemset of a pattern with a
varying length m ∈ [max]. Line 7 evaluates the concrete length of
the candidate pattern. The (choice) rule in line 5 allows for picking
a subset of Σ per slot to build a pattern. As a result, the rules from
lines 1 to 5 establish a unique representation for a candidate pattern
whose frequency remains to be checked.
To this end, the rules from Line 9 to 12 traverse each sequence
t = ht1 . . . tn i in D to derive atoms of the form emb(t,p,x)
which represent all the embeddings of the pattern in a sequence. As
noted in Table 1, for p ∈ [n] and x ∈ [max], such an atom expresses that it exists (i )i∈[p−1] such that pi ⊆ ti for all i ∈ [p − 1]
and px ⊆ tp , where p ∈ [n] and x ∈ [m]. Thus, it also means that
hp1 . . . px i v ht1 . . . tp i. Line 9 locates the occurrences of the first
itemset in the sequence. The expression seq(t,p,e):pat(1,e)
ensures that every e ∈ p1 must hold at position p of the sequence
t. Line 10 finds out locations for px , the x-th itemset of the pattern, from a position q such that q ≥ i for all i ∈ [p − 1],
where (i )i∈[p−1] is an embedding of hp1 . . . px−1 . The atoms
emb(t,p,x) are further inspected, line 14, to determine the cover
of a candidate pattern p. In these lines, the sequence is covered when
a complete embedding has been built, i.e. up to the length of the
pattern. The frequency of p is then checked in Line 16, where the
cardinality constraint “k { cover(t) }” over atoms cover(t),
signaling s v t, expresses that at least the frequency threshold k
many sequences t must include s.
Example 1. For the database in Listing 1, the pattern ha (bc)i leads
to atoms emb(3,1,1), emb(3,1,3) and emb(3,2,2) among
which emb(3,2,2) in turn yields cover(3) for the sequence
ha (bc) ai denoted by 3. There is not any sequence for which a
complete embedding can be built, and ha (bc)i turns out to not be
frequent, given the threshold k = 2.

4

Negative sequential patterns

In this section, we introduce the notion of negative sequential pattern
by giving their syntax and semantics borrowed from [8].

Table 1.

Atoms representing a pattern p = hpi i1≤i≤m , where 1 ≤ m ≤ max, and a given multiset D of sequences t = htj i1≤j≤n

Atom
slot(x)
pat(x,e)
item(e)
cover(t)
emb(t,p,x)

Meaning
1 ≤ x ≤ m refers to the position x of an item sx in p
px = e is the item at position x in p, where x ∈ [m]
item e belongs to some t ∈ D
hp1 . . . pm i v ht1 . . . tn i, that is, p v t
∃(ei )i∈[p−1] such that pi ⊆ tei for all i ∈ [p − 1] and px ⊆ tp , where p ∈ [n] and x ∈ [m].

Definition 1 (Negative sequential patterns (NSP)). A negative sequential pattern p = hp1 ¬q1 p2 ¬q2 · · · pn−1 ¬qn−1 pn i is a
finite sequence where pi ∈ 2Σ \ {∅} for all i ∈ [n] and qi ∈ 2Σ for
all i ∈ [n − 1].
p+ = hp1 . . . pn i is called the positive part of the NSP.
It can be noticed that Definition 1 introduces syntactic limitations
on negative sequential patterns that are commonly encountered in the
state of the art [17]: 1) a pattern can neither start or finish by a negative itemset, 2) a pattern cannot have two successive negative itemsets. But a pattern can have successive positive itemsets considering
that a negative itemsets can be empty.

Table 2 compares the support of patterns under the two semantics
of itemset non-inclusion. Since the positive part of p2 is in s2 , p2
occurs in the sequence iff (cd) 6⊆∗ (cf ). As for total non-inclusion,
it is false that (cd) 6⊆D (cf ) because c occurs in (cf ), and thus
p2 does not occur in s2 . As for partial non-inclusion, it is true that
(cd) 6⊆G (cf ), because d does not occur in (cf ), and thus p2 occurs
in s2 .
Table 2. Lists of sequences in D supported by negative patterns (pi )i=1..4
under the total and partial non-inclusion relations. Each pattern has the form
hb ¬qi ai where qi are itemsets such that qi ⊂ qi+1 .

Example 2. This example illustrates the notations introduced
in Definition 1. Consider Σ = {a, b, c, d} and p =
ha ¬(bc) (ad) d ¬(ab) di. Let p1 = {a}, p2 = {ad}, p3 = {d},
p4 = {d} and q1 = {bc}, q2 = ∅, q3 = {ab}. p+ = ha (ad) d di.

4.1

p1
p2
p3
p4

Semantics of negative sequential patterns

The semantics of negative sequential patterns relies upon negative
containment: a sequence s contains pattern p iff s contains a subsequence s0 such that every positive itemset of p is included in some
itemset of s0 in the same order and for any negative itemset ¬qi of
p, qi is not included in any itemset occurring in the sub-sequence of
s0 located between the occurrence of the positive itemset preceding
¬qi in p and the occurrence of the positive itemset following ¬qi in
p.
It turns out that the notion of negative containment has different
semantics depending on the definition of “not included” itemsets and
on the itemsets of the sequence to consider.
We introduce two relations comparing two itemsets P ∈ 2Σ \ {∅}
and I ∈ 2Σ :

= hb ¬c ai
= hb ¬(cd) ai
= hb ¬(cde) ai
= hb ¬(cdeg) ai

partial
non-inclusion
6⊆G

total
non-inclusion
6⊆D

{s1 , s3 , s4 }
{s1 , s2 , s3 , s4 }
{s1 , s2 , s3 , s4 }
{s1 , s2 , s3 , s4 , s5 }

{s1 , s3 , s4 }
{s1 , s4 }
{s1 }
{s1 }

Now, we formulate the notions of sub-sequence, non-inclusion and
absence by extending the concept of embedding (see Section 2.1) to
negative patterns.
Definition 2 (Strict and soft embeddings of negative patterns). Let
s = hs1 . . . sn i be a sequence and p = hp1 ¬q1 . . . ¬qm−1 pm i
be a negative sequential pattern.
An increasing3 tuple  = (i )i∈[m] ∈ [n]m is a soft-embedding of
pattern p in sequence s iff:

• partial non inclusion: P 6⊆G I ⇔ ∃e ∈ P , e ∈
/I
• total non inclusion: P 6⊆D I ⇔ ∀e ∈ P, e ∈
/I

• pi ⊆ sei for all i ∈ [m]
• qi ⊆
6 ∗ sj , for all j ∈ [ei + 1, ei+1 − 1] and for all i ∈ [m − 1]

Partial non-inclusion means that P \ I is non-empty while total noninclusion means that P and I are disjoint. By convention, ∅ 6⊆D I
and ∅ 6⊆G I for all I ⊆ Σ.
In the sequel, we denote the general form of itemset non-inclusion
by the symbol 6⊆∗ , meaning either 6⊆G or 6⊆D .
Intuitively, partial non-inclusion identifies the itemset P with a
disjunction of negative constraints, i.e. at least one of the items (of P )
has to be absent from I, and total non-inclusion consider the itemset
P as a conjunction of negative constraints: all items (of P ) have to
be absent from I.
Choosing one non-inclusion interpretation or the other has consequences on extracted patterns as well as on pattern search. Let us
illustrate this with the following dataset of sequences:


s1 = h(bc) f ai




 s2 = h(bc) (cf ) ai



s3 = h(bc) (df ) ai
D=
.




 s4 = h(bc) (ef ) ai



s5 = h(bc) (cdef ) ai

An increasing3 tuple e = (ei )i∈[m] ∈ [n]m is a strict-embedding
of pattern p in sequence s iff:
• pi ⊆ seSi for all i ∈ [m]
• qi ⊆
6 ∗ j∈[ei +1,ei+1 −1] sj for all i ∈ [m − 1]
Intuitively, the constraint of a negative itemset qi is checked on
the sequence’s itemsets at positions in interval [ei + 1, ei+1 − 1],
i.e. between occurrences of the two positive itemsets surrounding the
negative itemset in the pattern. A soft embedding considers individually each of the sequence’s itemsets of [ei + 1, ei+1 − 1] while a
strict embedding consider them as a whole.
Example 3 (Itemset absence semantics). Let p = ha ¬(bc) di be a
pattern and consider four sequences as follows:
3

By an increasing tuple e, we mean a tuple such that i < i+1 (in particular,
repetitions are not allowed).

Sequence
s1
s2
s3
s4

= ha c b e di
= ha (bc) e di
= ha b e di
= ha e di

6 D
⊆
strict

6⊆D
soft

6 G
⊆
strict

The strong containment relation states that a negative pattern p occurs in a sequence s iff there exists at least one occurrence of the positive part of pattern p in sequence s and every such occurrence satisfies the negative constraints; the strong containment relation states
p occurs in a sequence s iff there exists at least one occurrence of
the positive part of pattern p in sequence s and at least one of these
occurrences satisfies the negative constraints.

6⊆G
soft
3

3

3

3
3

3
3

Notice that each sequence contains a unique occurrence of p+ =
ha di, the positive part of pattern p. Considering soft-embedding and
partial non-inclusion (6⊆∗ :=6⊆G ), p occurs in s1 , s3 and s4 but not
in s2 . Considering strict-embedding and partial non-inclusion, p occurs in s3 and s4 . Indeed, items b and c occur between occurrences
of a and d in s1 and s2 . Considering total non-inclusion (6⊆∗ :=6⊆D )
and either type of embeddings, the absence of an itemset is satisfied
if any of its items is absent. Hence, p occurs only in s4 .
Let us now give some lemmata about the relations between the
different types of embedding.4
Lemma 1. In the case that 6⊆∗ is 6⊆D , e is a soft-embedding iff e is
a strict-embedding
Lemma 2. Let p = hp1 ¬q1 . . . ¬qn−1 pn i ∈ N such that |qi | ≤
1 for all i ∈ [n − 1], then e is a soft-embedding iff e is a strictembedding.
At this point, we have exhibited four semantics for negative pattern embeddings. Lemmata 1 and 2 concluded to the equivalence of
soft and strict-embeddings when the itemset non-inclusion relation is
6⊆D .
In [8], we proposed notations to disambiguate syntactically the
different types of negation (¬(a1 , . . . , ali ) or ¬{a1 , . . . , ali } or
¬|a1 , . . . , ali |). ]l, u[ denotes the interval of integers [l + 1, u − 1].
• ¬(a1 , . . . , ali ) is evaluated as
{a1 , . . . , ali } 6⊆G sj , ∀j ∈]ei , ei+1 [ for all i ∈ [m − 1] (2)
Intuitively, you check that, in between sei (i.e., a match for pi )
and sei+1 (i.e., a match for pi+1 ), none of these sj include all of
a1 , . . . , ali .
• ¬{a1 , . . . , ali } is evaluated as[
{a1 , . . . , ali } 6⊆G

sj for all i ∈ [m − 1]

(3)

j∈]ei ,ei+1 [

Intuitively, we check that there exists some item in a1 , . . . , ali that
does not occur at all in between sei (i.e., a match for pi ) and sei+1
(i.e., a match for pi+1 ).
• ¬|a1 , . . . , ali | is evaluated as
{a1 , . . . , ali } 6⊆D sj , ∀j ∈]ei , ei+1 [ for all i ∈ [m − 1] (4)
Intuitively, we check that every item in a1 , . . . , ali fails to occur
in between sei (i.e., a match for pi ) and sei+1 (i.e., a match for
pi+1 ).

4.2

Multiple occurrences

Another point that determines the semantics of negative containment
concerns the multiple occurrences of some pattern in a sequence:
should at least one or should all occurrences of the pattern positive
part in the sequence satisfy the non-inclusion constraints?
4

Proofs of these Lemmata can be found in the extended version of [8].

Example 4 (Strong vs weak occurrence semantics). Let p =
ha b ¬c di be a pattern, s1 = ha b e di and s2 = ha b c a d e b di
be two sequences. Thus, p+ = ha b di occurs once in s1 hence there
is no difference for occurrences of p in s1 under the two semantics.
However, p+ occurs four times in s2 through embeddings (1, 2, 5),
(1, 2, 8), (1, 7, 8) and (4, 7, 8). The first two occurrences do not satisfy the negative constraint (¬c) but the last two occurrences do. Under the weak occurrence semantics, pattern p occurs in sequence s2
whereas it fails to do so under the strong occurrence semantics.

5

ASP encodings of negative sequential patterns

In the following, we introduce new facts to encode negations in negative sequential patterns. The pat(x,e) atom now represents the
positive part of a negative sequential pattern. The negated itemsets
are encoded by terms of three predicates: negpat/2 (for ¬| · |),
cnegpat/2 (for ¬{·}) and csnegpat/2 (for ¬(·)). For instance,
the pattern ha¬|b|a¬(bd)ci is encoded by the facts pat(1,a),
pat(2,a), pat(3,c), negpat(1,b), csnegpat(2,b) and
csnegpat(2,d).
The following choice rule illustrates the generation of negative
itemsets of kind ¬| · |. Similar rules yield the negative part of a candidate pattern for the two other kinds of negations.
{ negpat(X,E): item(E), not pat(X,E) } :slot(X), X<L, patlen(L).
Note that we assume that it is not possible to have several different
kinds of negated itemsets at the same position. For instance, pattern
ha¬|b|¬{c}di is not possible.
In the next section, we give the rules for negative constraints. A
negative constraint denotes a constraint upon an embedding (i )i∈[n]
of the positive part that is expressed by a negated itemset in the candidate negative pattern. If all negative constraints are satisfied, then
(i )i∈[n] is an embedding of the negative pattern. Then, it is further
used to evaluate the containment of the candidate pattern p in a sequence t.
The core of the encoding relies on the construction of an embedding. We adapt the definition of the predicate emb/3 to negative
sequential patterns as follow: let p = hp1 ¬q1 . . . pm i be a negative sequential pattern, t = ht1 . . . tn i be a sequence, and ν a
Boolean value, then emb(t,ρ,x,ν) means that ∃(ei )i∈[x−1] such
that x ∈ [m], ρ ∈ [n] and
1. pi ⊆ tei for all i ∈ [x − 1],
2. px ⊆ tρ ,
3. ∃y ∈ [x − 1] s.t. qy is not satisfied for the partial embedding
(ei )i∈[p−1] if ν is true,
4. ∀y ∈ [x − 1] s.t. qy is satisfied for the partial embedding
(ei )i∈[p−1] if ν is false,
Note that “qy is not satisfied for the partial embedding (ei )i∈[p−1] ”
refers to the semantics of negation (see previous section) whatever
the kind of occurrence (weak or strong).

1 emb(T,P,1,0)

:- seq(T,P,E):pat(1,E); seq(T,P,_).

3 emb(T,P,X+1,1) :- emb(T,Q,X,1); Q<P;
4
seq(T,P,E):pat(X+1,E); seq(T,P,_);
5
patlen(L); X<L.
7 emb(T,P,X+1,0) :- emb(T,Q,X,0);
8
seq(T,P,E):pat(X+1,E); seq(T,P,_); P>Q;
9
#count{ R : not seq(T,R,Ep), csnegpat(X,Ep), R=(Q+1)..(P-1)}=(P-Q-1);
10
patlen(L); X<L.
12 emb(T,P,X+1,1) :- emb(T,Q,X,0);
13
seq(T,P,E):pat(X+1,E); seq(T,P,_); P>Q;
14
#count{ R : not seq(T,R,Ep), csnegpat(X,Ep), R=(Q+1)..(P-1)}!=(P-Q-1);
15
patlen(L); X<L.
Listing 3. Encoding for embedding with ¬(·)/csnegpat/2 negative itemsets

1 emb(T,P,X+1,1) :- emb(T,Q,X,0); Q<P;
2
seq(T,P,E):pat(X+1,E); seq(T,P,_);
3
#count{ R : seq(T,R,Ep), negpat(X,Ep), Q<R, R<P}!=0;
4
patlen(L); X<L.
6 emb(T,P,X+1,0) :- emb(T,Q,X,0); Q<P;
7
seq(T,P,E):pat(X+1,E); seq(T,P,_);
8
#count{ R : seq(T,R,Ep), negpat(X,Ep), Q<R, R<P}=0;
9
patlen(L); X<L.
Listing 4. Encoding for embedding with ¬| · |/negpat/2 negative itemsets

ν is a Boolean value that stores the information for an embedding
of the positive part that these embedding satisfies or not the negative
constraints.
With this definition it happens that if there is an atom of the embedding emb(t,p,l,ν) where l is the size of the positive part of
the pattern whatever the ν value, then the positive part occurs in the
sequence t. If ν is false, this means that there exists an embedding for
the negative sequential pattern, thus p weakly occurs in t. If there is
no embedding such that ν is true, then it means that no embedding of
the positive part is an embedding of the negative sequential pattern.
Thus p strongly occurs in t if it is not the case. The following rules
encode the rules of pattern covering for the two different definitions
of weak and strong occurrences of a candidate pattern.
weakcover(T) :- patlen(L); emb(T,_,L,0).
strongcover(T) :- patlen(L); emb(T,_,L,_);
#count{ T : emb(T,_,L,1) }=0.
Then, depending on the kind of covering considered, line 16 of
Listing 2 is adapted to the minimal occurrence number constraint.
The remainder of this section details the encoding for each semantics of negative itemsets.

5.1

Encoding of ¬(·) negative constraints

Listing 3 gives the encoding yielding the emb/4 atoms considering
patterns with ¬(·)/csnegpat/2 negative itemsets.
In this encoding, the Line 1 states that 1 of an embedding is a position p in the sequence t such p1 ⊆ tp . For the first pattern itemset,
there are no negative constraints at previous position. Thus, ν is false
(0).
The rule in lines 3-5 is similar to the rule in lines 10-12 of Listing 2. They yield the x+1 of an embedding of the positive part of
the pattern where (i )i∈[x] is an embedding of hp1 ¬q1 . . . px i s.t.

there exists y ∈ [x − 1] s.t. qy is not satisfied (ν is true). In this
case, (i )i∈[x+1] is an embedding of hp1 ¬q1 . . . px+1 i that neither satisfies all the negative constraints. Thus, the truth value of ν is
propagated.
The rules in lines 7-15 consider the case of an embedding (i )i∈[x]
of the negative pattern hp1 ¬q1 . . . px i such that all the negative
constraints are satisfied. Then, there are two cases for the embedding
(i )i∈[x+1] . If there exists a position p in the sequence t such that
px+1 ⊆ tp and the negative constraint qx is satisfied, then the atom
emb(t,p,l,ν) is yielded with ν set to false (0). That is, the rule in
lines 7-10. But, if there exists a position p, p > x in the sequence
t such that px+1 ⊆ tp and the negative constraint qx is not satisfied, then the atom emb(t,p,l,ν) is yielded with ν set to true (1).
That is, the rule in lines 12-15. Note that these two rules could be
factorised to avoid redundancies. For sake of readability, we prefer
to present this redundant syntax.
Let us now detail how to evaluate the negative constraint qx (i.e.,
the set of atoms csnegpat(x,e)). According to Equation 2, the
negative constraint qx is satisfied iff ∀r ∈]i , i+1 [, ∃e ∈ qx , s.t. e 6∈
tr . It is equivalent to evaluate whether |{r ∈]i , i+1 [|∃e ∈ qx , e 6∈
tr }| is equal to i+1 − i + 1. This latter constraint is encoded line 9
and its opposite is encoded line 14. The #count is an aggregate to
compute the number of different R values satisfying conditions after
the colon.

5.2

Encoding of ¬| · | negative constraints

The negative constraints qx with ¬| · |/negpat/2 is satisfied iff
∀r ∈]i , i+1 [, ∀e ∈ qx , e 6∈ tr (see Equation 4). This means that it
is not satisfied iff {r ∈]i , i+1 [|e ∈ qx ∧ e ∈ tr } is empty.
Listing 4 yields atoms of predicate emb/4 with the same principle
of Listing 3. The lines 3 or 8 evaluate |{r ∈]i , i+1 [|e ∈ qx ∧ ∈

1 negislen(X,N) :- X=1..(L-1); patlen(L); N=#count{E : cnegpat(X,E)}.
3 emb(T,P,X+1,1) :- emb(T,Q,X,0); Q<P;
4
seq(T,P,E):pat(X+1,E); seq(T,P,_);
5
negislen(X,N);
6
N=#count{ Ep : seq(T,R,Ep), cnegpat(X,Ep), Q<R, R<P};
7
patlen(L); X<L.
9 emb(T,P,X+1,0) :- emb(T,Q,X,0); Q<P;
10
seq(T,P,E):pat(X+1,E); seq(T,P,_);
11
negislen(X,N);
12
N!=#count{ Ep : seq(T,R,Ep), cnegpat(X,Ep), Q<R, R<P};
13
patlen(L); X<L.
Listing 5. Encoding for embedding with ¬{·}/cnegpat/2 negative itemsets

tr }|. It is compared to 0 to assign the truth value to ν for x+1 (if
exists, i.e., px+1 ⊆ tx+1 ).

5.3

Encoding of ¬{·} negative constraints

Finally, the negative constraint
qx with ¬{·}/cnegpat/2 is satisS
fied iff ∃e ∈ qx , e 6∈ r∈]i ,i+1 [ tr (see Equation 3). This means
S
that it is not satisfied iff ∀e ∈ qx , e ∈ r∈]i ,i+1 [ tr or in other
S
words that |{e ∈ qx |e ∈ r∈]i ,i+1 [ tr }| = |qx−1 |.
Listing 5 yields atoms of predicate emb/4 with the same principle of Listing 3. Line 1 evaluates the size of negative itemsets.
negislen(x,n)
S states that |qx | = n. Lines 6 and 12 evaluate
|{e ∈ qx |e ∈ r∈]i ,i+1 [ tr }|. It is compared to |qx | to assign the
truth value to ν for x+1 (if exists, i.e., px+1 ⊆ tx+1 ).

memory and ran with a single thread. A timeout was set up to 5 minutes.
For sake of better interpretability of the results and of computation time, the negative sequential patterns contain only one kind of
negated itemset (¬| · |, ¬(·) or ¬{·}), the length of the positive part
of a pattern is at most 3, and the size of negated itemsets is limited to
2 items.
Note that we do not compare our approach with dedicated algorithms, e.g. [10, 17] that are undoubtly significantly more efficient for
this task. We remind that declarative pattern mining aims at achieving
reasonable computation time for mid-sized set of examples. Thus, it
can be used to inject some expert knowledge in the mining process
to reduce the number of patterns.

6.2
6

Experiments and results

The previous section exhibited the ease of modeling of the different
semantics of negative sequential patterns in order to extract the frequent ones. It is worth noticing that only few of these semantics have
properties to design algorithms that are both efficient and complete
[8]. More precisely, the anti-monotonicity property holds with weak
occurrences of ¬| · |, and weaker versions of anti-monotonicity hold
with the strong occurrences of ¬| · |.
The question investigated in these experiments is whether the
solver takes advantage of these properties to achieve the mining task
within a lower computation time.

6.1

Experimental setting

We conducted experiments on synthetic sets of sequences generated
with the generator of [2], without hidden patterns (pure random sequences).
The usage of synthetic sets of sequences allows us to control and
analyze the most important characteristics of data wrt the computing
time. To be more precise, we randomly generated sets of 100 sequences. Each sequence contains 10 itemsets. Each itemset is of size
2. Items are randomly generated according to a Gaussian law (some
items are more frequent than others) over a vocabulary of 20 items.
For each data point, we then give average computing times over five
similar sets of sequences.
All the experiments have been using the clingo solver5 on a desktop computer with enough memory to prevent from using cache
5

clingo version 5.3.1

Results

Figure 1 compares the number of patterns that are extracted through
the different semantics of negative patterns. Note that it is very interesting to have this study as no algorithm can extract a complete set
of frequent negative sequential patterns for all negation semantics.
The figures take into account only the number of patterns for runs
ended before timeout (5 minutes). It explains that there are only few
results for ¬{·} which is often out of the timeout.
The results illustrated in Figure 1 are consistent with theoretical
results [8]. First, we notice that there are more patterns extracted
with weak-occurrences than with strong-occurrences. For identical
settings, there are in average 2.80 ± 2.13 times more patterns with
weak-occurrences than with strong-occurrences. Second, ¬| · | is the
strongest negative constraint after ¬{·} and finally ¬(·) is the weakest. As a consequence, there are more extracted patterns for ¬(·),
than for ¬| · |. The number of patterns for ¬{·} is in-between when
available.
Figure 2 compares computing times of ASP-based negative sequential pattern mining with different semantics of negative patterns.
We can first notice that extracting negative sequential patterns with
¬{·} lacks efficiency and the mining tasks does not finish before
timeout in most cases.
Then, despite the larger number of occurrences of weakoccurrences, evaluating weak occurrences is more efficient that evaluating the strong ones. It is unsurprising. Indeed, as soon as a weakoccurrence has been found, the atom cover(t) is proved to be true
without necessarily exploring the other embeddings of the positive
part of the pattern. For strong occurrences, it is mandatory to evaluate negative patterns for every embedding of the positive part of the
candidate pattern.

Figure 1. Boxplots of the numbers of patterns for computing all frequent negative sequential patterns wrt different thresholds (24, 27 and 30) using ASP. Each
plot corresponds to one specific semantic, from left to right ¬{·}, ¬| · | and ¬(·). Blue (resp. red) boxplots correspond to weak (resp. strong) occurrences).

Figure 2. Boxplots of times for computing all frequent negative sequential patterns wrt different thresholds (24, 27 and 30) using ASP. Each plot corresponds
to one specific semantic, from left to right ¬{·}, ¬| · | and ¬(·). Blue (resp. red) boxplots correspond to weak (resp. strong) occurrences).

The surprising result of this experiment is in the comparison between ¬(·) and ¬| · |. It turns out that mining negative sequential
patterns with ¬(·) is significantly more efficient than with ¬| · |. This
result is enhanced by the fact that there are many more patterns to
extract with ¬(·) than with ¬| · |. So there are more patterns but more
efficiently extracted. The opposite outcome was expected due to our
theoretical results that exhibited anti-monotonicity for ¬| · |, but not
for ¬(·).

7

Conclusion

This article extends the framework of mining sequential patterns
with Answer Set Programming to negative sequential patterns. We
encoded the three different semantics of absent events proposed in
[8]. The first interest of the approach is to evaluate the impact of the
choice of one of the semantics on the set of extracted patterns. In
this article, we focused on the size of this set. Deeper experiments on
real datasets would be interesting to identify the differences between
these sets of patterns.
Our experiments on synthetic data show that the kind of negated
itemset for which the encoding is the most efficient is not the expected one. Indeed, we expected to have better efficiency with ¬| · |.
A possible explanation of this observation may be that there are also
anti-monotonicity properties of ¬(·) for partial orders that are not
usually investigated for pattern mining algorithms [10]. More inves-

tigation is required to confirm this intuition. But it encourages us to
think that there are potentially better encodings for ¬| · |. One of our
perspectives is to investigate saturation programming techniques that
may be suitable and efficient to encode weak-occurrences of negative
sequential patterns.
In addition, these encodings may be used to characterize what patterns can still be mined by incomplete mining heuristics. Wang et al.
[17] reviewed such algorithms but they do not characterise the extracted patterns among the complete set of patterns. Comparing their
outputs with the outputs of our encodings makes it possible to investigate this question.

